
Macro 1: 
Effect size ratio method for sample 
size re-estimation

%Macro es(nsim=1000000,alpha=0.025, beta=0.1, sigma=50, ni=0, rx=150, 
 ry=150,n1=70, nmax=350, n0=233, diff=15,alpha1=0.0026, alpha0=1, 

alpha2=0.0240);
data es;keep FSP ESP nmean power nclassic; seedx=1736; seedy=6214;alpha=&alpha; ni=&ni;
 nmax=&nmax; rx=&rx; ry=&ry; sigma=&sigma;n1=&n1;
 delta= abs(&diff+ni)/sigma;
 nclassic=round(2*((probit(1-alpha)+probit(1-&beta))/delta)**2);
 FSP=0; ESP=0; nmean=0;power=0;
 Do isim= 1 to &nsim;
  rx1=rannor(seedx)*sigma/sqrt(n1)+rx;
  ry1=rannor(seedy)*sigma/sqrt(n1)+ry;
  t1=(rx1-ry1+ni)*sqrt(n1)/2**0.5/sigma;
  p1=1-probnorm(t1);
  if p1>&alpha0 then do;
   FSP=FSP+1/&nsim; nfinal=n1;
  end;
  if p1<=&alpha1 then do;
   power=power+1/&nsim; ESP=ESP+1/&nsim; nfinal=n1;
  end;
  if p1>&alpha1 and p1<=&alpha0 then do;
   if &diff*(rx1-ry1+ni)<0 then nfinal=n1;
   er=&diff/(abs(rx1-ry1)+0.0000001);
   nfinal=min(&nmax, max(&n0, er**2*&n0));
   if nfinal>n1 then do;
    rx2=rannor(seedx)*sigma/sqrt(nfinal-n1)+rx;
    ry2=rannor(seedy)*sigma/sqrt(nfinal-n1)+ry;
    t2=(rx2-ry2+ni)*sqrt(nfinal-n1)/2**0.5/sigma;
    z2=(t1+t2)/sqrt(2);
    p2=1-probnorm(z2);
    if .<p2<=&alpha2 then power=power+1/&nsim;
   end;
  end;
  nmean=nmean+nfinal/&nsim;
 end;
 proc print data=es; run;
%Mend es;

Macro 2: 
Conditional power method for sample 
size re-estimation
%Macro cp(nsim=1000000,alpha=0.025, beta=0.1, sigma=50, ni=0, rx=150, 

 ry=150,n1=70,nmax=280,diff=15,alpha1=0.0026,alpha0=1,alpha2=0.0240, 
w=0.70711,cP=0.9);

data n;keep FSP ESP nmean power nclassic;
 seedx=1736; seedy=6214; alpha=&alpha; ni=&ni; rx=&rx; ry=&ry; sigma=&sigma; 

n1=&n1;
 delta=abs(&diff+ni)/sigma;
 nclassic=round(2*((probit(1-alpha)+ probit(1-&beta))/delta)**2);
 FSP=0; ESP=0; nmean=0;power=0;
 Do isim= 1 to &nsim;
  rx1=rannor(seedx)*sigma/sqrt(n1)+rx;
  ry1=rannor(seedy)*sigma/sqrt(n1)+ry;
  t1=(rx1-ry1+ni)*sqrt(n1)/2**0.5/sigma;
  p1=1-probnorm(t1);
  if p1>&alpha0 then do;
   FSP=FSP+1/&nsim; n2=0;
  end;
  if p1<=&alpha1 then do;
   power=power+1/&nsim; ESP=ESP+1/&nsim; n2=0;
  end;
  if p1>&alpha1 and p1<=&alpha0 then do;
   eSize=&diff/&sigma;
   Cfun=(probit(1-&alpha2)-&w*probit(1-p1))/&w;
   n2=min(&nmax,2*((Cfun-probit(1-&cP))/eSize)**2);
   rx2=rannor(seedx)*sigma/sqrt(n2)+rx;
   ry2=rannor(seedy)*sigma/sqrt(n2)+ry;
   t2=(rx2-ry2+ni)*sqrt(n2)/2**0.5/sigma;
   z2=(t1+t2)/sqrt(2);
   p2=1-probnorm(z2);
   if .<p2<=&alpha2 then power=power+1/&nsim;
  end;
  nmean=nmean+(n1+n2)/&nsim;
 end;
 proc print data=n; run;
%Mend cp;

